

NAG Library Routine Document

F08JFF (DSTERF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08JFF (DSTERF) computes all the eigenvalues of a real symmetric tridiagonal matrix.

2 Specification

```
SUBROUTINE F08JFF(N, D, E, INFO)
  INTEGER          N, INFO
  double precision D(*), E(*)
```

The routine may be called by its LAPACK name *dsterf*.

3 Description

F08JFF (DSTERF) computes all the eigenvalues of a real symmetric tridiagonal matrix, using a square-root-free variant of the *QR* algorithm.

The routine uses an explicit shift, and, like F08JEF (DSTEQR), switches between the *QR* and *QL* variants in order to handle graded matrices effectively (see Greenbaum and Dongarra (1980)).

4 References

Greenbaum A and Dongarra J J (1980) Experiments with QR/QL methods for the symmetric triangular eigenproblem *LAPACK Working Note No. 17 (Technical Report CS-89-92)* University of Tennessee, Knoxville

Parlett B N (1998) *The Symmetric Eigenvalue Problem* SIAM, Philadelphia

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix T .
Constraint: $N \geq 0$.
- 2: D(*) – **double precision** array *Input/Output*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: the diagonal elements of the tridiagonal matrix T .
On exit: the n eigenvalues in ascending order, unless $\text{INFO} > 0$ (in which case see Section 6).
- 3: E(*) – **double precision** array *Input/Output*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On entry: the off-diagonal elements of the tridiagonal matrix T .
On exit: E is overwritten.
- 4: INFO – INTEGER *Output*
On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm has failed to find all the eigenvalues after a total of $30 \times N$ iterations. If INFO = i , then on exit i elements of E have not converged to zero.

7 Accuracy

The computed eigenvalues are exact for a nearby matrix $(T + E)$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the *machine precision*.

If λ_i is an exact eigenvalue and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq c(n)\epsilon\|T\|_2,$$

where $c(n)$ is a modestly increasing function of n .

8 Further Comments

The total number of floating-point operations is typically about $14n^2$, but depends on how rapidly the algorithm converges. The operations are all performed in scalar mode.

There is no complex analogue of this routine.

9 Example

This example computes all the eigenvalues of the symmetric tridiagonal matrix T , where

$$T = \begin{pmatrix} -6.99 & -0.44 & 0.00 & 0.00 \\ -0.44 & 7.92 & -2.63 & 0.00 \\ 0.00 & -2.63 & 2.34 & -1.18 \\ 0.00 & 0.00 & -1.18 & 0.32 \end{pmatrix}.$$

9.1 Program Text

```
*      F08JFF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*      .. Local Scalars ..
      INTEGER          I, INFO, N
*      .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), E(NMAX-1)
*      .. External Subroutines ..
      EXTERNAL        DSTERF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08JFF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read T from data file
*
```

```

      READ (NIN,*) (D(I),I=1,N)
      READ (NIN,*) (E(I),I=1,N-1)
*
*      Calculate the eigenvalues of T
*
      CALL DSTERF(N,D,E,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.GT.0) THEN
        WRITE (NOUT,*) 'Failure to converge.'
      ELSE
        WRITE (NOUT,*) 'Eigenvalues'
        WRITE (NOUT,99999) (D(I),I=1,N)
      END IF
    END IF
*
99999 FORMAT (3X,(9F8.4))
      END

```

9.2 Program Data

F08JFF Example Program Data

```

      4                               :Value of N
      -6.99   7.92   2.34   0.32
      -0.44  -2.63  -1.18           :End of matrix T

```

9.3 Program Results

F08JFF Example Program Results

```

Eigenvalues
      -7.0037 -0.4059  2.0028  8.9968

```
